
MATCH関数
使い⽅マスターガイド

※当ガイドに従うことで、法令違反がないことを保証する資料ではありません。
※あくまで参考としてご活⽤いただくことを想定している資料です。実際の制度内容は国の資料等をご確認ください。

MATCH関数の「役割」を理解する

MATCH関数は、「範囲の中で、探したいデータが上から（左から）何番⽬にあるか」という「位置（番号）」だけを返す関数
です。

=MATCH(検索値, 検索範囲, [照合の種類])

引数 内容

① 検索値 何を探すか。（例: "商品B"、1005、B2セル）

② 検索範囲 どこから探すか。（例: A2:A10）※1列または1⾏のみ

③ 照合の種類
どう探すか。（0=完全⼀致、1=以下、-1=以上）

実務では、ほぼ100%「0 (完全⼀致)」を使います。

💡例：  =MATCH("商品B", A2:A10, 0) → A2:A10の範囲で「商品 B」は3番目にある場合、 3 という「数値」が返されます。

VLOOKUPの「弱点」と「INDEX + MATCH」

MATCH関数の真価は、VLOOKUPの弱点を克服することにあります。

VLOOKUPの弱点

検索値は、必ず範囲の「一番左」の列になければならない。

（例：社員番号（ A列）で氏名（B列）を探すことはできるが、氏名（ B列）で社員番号（ A列）を探すことはできない）

INDEX + MATCHという最強タッグ

この弱点を、 INDEX関数とMATCH関数を組み合わせることで解決します。

MATCH関数 → 「行（列）の番号」を見つける（検索役）

INDEX関数 → 「指定された範囲」から MATCH が返した「◯番目」のデータを取り出す（抽出役）



【実践】「INDEX + MATCH」の組み⽴て⽅



【実践】「INDEX + MATCH」の組み⽴て⽅

この2つを以下のように「合体」させます。

=INDEX( [取り出したいデータの列] , MATCH( [検索値] , [検索値がある列] , 0) )

シナリオ：  社員名簿（D列）から「田中」さんを探し、対応する社員番号（ B列）を取り出したい。  
　　　　　（VLOOKUPでは検索できない左側検索のパターン）

数式：  =INDEX( B:B , MATCH("田中", D:D, 0) )

解説：

1. MATCH("田中", D:D, 0) 

→D列（社員名簿）から「田中」さんを探し、その「位置（行番号）」を見つけます。（例 : 5）

2. INDEX(B:B, 5) 

→B列（社員番号）の「 5番目」のデータを取り出します。これで、検索列が左端になくても自由にデータを抽出できます。



その他の応⽤テクニック



その他の応⽤テクニック

MATCH関数は、「列番号」を動的に指定するためにも使われます。

組み合わせ 内容と数式例

VLOOKUP + MATCH

【⽬的】VLOOKUPの「列番号」を可変にする

例： VLOOKUPの3番⽬の引数（列番号）を 3 や 4 と⼿⼊⼒する代わりに、MATCH関数で
「⾒出し」を検索させて⾃動で指定する。
=VLOOKUP(A10, A1:E5, MATCH(B9, A1:E1, 0), FALSE)

HLOOKUP + MATCH

【⽬的】HLOOKUPの「⾏番号」を可変にする

上記VLOOKUP検索範囲が横から縦へ逆になったパターン。HLOOKUPの3番⽬の引数（⾏番
号）をMATCH関数で指定する。
=HLOOKUP(A10, A1:E5, MATCH(B9, A1:A5, 0), FALSE)

INDEX + MATCH + MATCH

【⽬的】⾏も列も「動的な表」から抽出する

縦軸（⾏）と横軸（列）の両⽅をMATCH関数で検索し、INDEXで「X座標とY座標」が交差
するセルの値をピンポイントで抽出する。
=INDEX(A1:E5, MATCH(A10, A1:A5, 0), MATCH(B9, A1:E1, 0))



【免責】

※当資料に従うことで、法令違反がないことを保証する資料ではありません。

※あくまで参考としてご活用いただくことを想定している資料です。また当資料は、公開日時点の内

容となっております。最新の情報、実際の制度内容は国の資料等をご確認ください。


